
各位同学：

这个程序来自 BLAS (Basic Linear Algebra Subprograms)，通过读这些源代码了解一个优秀的程序
是如何编的。在这门课程中，我用这些例子强调一个概念，即在学习编程的时候，培养一个良好的

编程习惯是非常重要的。

几个重要的习惯如下：

1. 要有写程序功能描写的习惯；
2. 对参数的定义做详细介绍；
3. 保留足够多的空格；
4. 对齐，让整个程序看起来整齐、划一；
5. 作者的信息，比如名字/单位等；
6. 参考资料和文献；
7. 程序编写时间；
8. 检查输入文件的格式，检查数据的合法性；
9. 丰富程序功能，使之有更加广泛的用途；

比如 IF (N.LE.0) RETURN
这个命令检验 N的范围；如果 N <=0, 表示输入参数是不合理的，因此程序不能执行。对于一个好
的子程序，应该有一定的智能性，即能够检查输入参数的合理性。这种合理性有很多，举例如下：

➢ A/x： 要求检查 x是否为零；
➢ N： 要求检查整数 N的范围和符号；对于数组，N>=1；
➢ ArcTan[x]: 要求|x| <= Pi/2;
➢ Exp[-x]: 要求 x >=0，或者 x不能为负无穷大；

下面分析这个程序的功能：

COMPLEX FUNCTION CDOTC(N,CX,INCX,CY,INCY)

CX, CY: 两个矢量:
INCX, INCY: 内积求和时的增量。

这个函数计算:

Cdotc = Sum_{a=1}^N CX^* (1 + (a-1) * INCX) * CX(1 + (a-1) * INCY)

可见，作者将矢量乘积做了推广。

注意：对于编程习惯，没有统一的标准；每个人都有自己的习惯。但是整齐、简洁、漂亮，是提高

程序可读性的关键。请各位同学根据习惯培养适合自己的编程习惯，请向最好的代码学习。

龚明/LQCC，计算物理。



*> \brief \b CDOTC
*
* =========== DOCUMENTATION ===========
*
* Online html documentation available at
* http://www.netlib.org/lapack/explore-html/
*
* Definition:
* ===========
*
* COMPLEX FUNCTION CDOTC(N,CX,INCX,CY,INCY)
*
* .. Scalar Arguments ..
* INTEGER INCX,INCY,N
* ..
* .. Array Arguments ..
* COMPLEX CX(*),CY(*)
* ..
*
*
*> \par Purpose:
* =============
*>
*> \verbatim
*>
*> CDOTC forms the dot product of two complex vectors
*> CDOTC = X^H * Y
*>
*> \endverbatim
*
* Arguments:
* ==========
*
*> \param[in] N
*> \verbatim
*> N is INTEGER
*> number of elements in input vector(s)
*> \endverbatim
*>
*> \param[in] CX
*> \verbatim
*> CX is COMPLEX array, dimension ( 1 + ( N - 1 )*abs( INCX ) )
*> \endverbatim
*>
*> \param[in] INCX
*> \verbatim
*> INCX is INTEGER
*> storage spacing between elements of CX
*> \endverbatim
*>
*> \param[in] CY
*> \verbatim
*> CY is COMPLEX array, dimension ( 1 + ( N - 1 )*abs( INCY ) )
*> \endverbatim
*>
*> \param[in] INCY



*> \verbatim
*> INCY is INTEGER
*> storage spacing between elements of CY
*> \endverbatim
*
* Authors:
* ========
*
*> \author Univ. of Tennessee
*> \author Univ. of California Berkeley
*> \author Univ. of Colorado Denver
*> \author NAG Ltd.
*
*> \ingroup complex_blas_level1
*
*> \par Further Details:
* =====================
*>
*> \verbatim
*>
*> jack dongarra, linpack, 3/11/78.
*> modified 12/3/93, array(1) declarations changed to array(*)
*> \endverbatim
*>
* =====================================================================

COMPLEX FUNCTION CDOTC(N,CX,INCX,CY,INCY)
*
* -- Reference BLAS level1 routine --
* -- Reference BLAS is a software package provided by Univ. of Tennessee, --
* -- Univ. of California Berkeley, Univ. of Colorado Denver and NAG Ltd..--
*
* .. Scalar Arguments ..

INTEGER INCX,INCY,N
* ..
* .. Array Arguments ..

COMPLEX CX(*),CY(*)
* ..
*
* =====================================================================
*
* .. Local Scalars ..

COMPLEX CTEMP
INTEGER I,IX,IY

* ..
* .. Intrinsic Functions ..

INTRINSIC CONJG
* ..

CTEMP = (0.0,0.0)
CDOTC = (0.0,0.0)
IF (N.LE.0) RETURN
IF (INCX.EQ.1 .AND. INCY.EQ.1) THEN

*
* code for both increments equal to 1
*

DO I = 1, N
CTEMP = CTEMP + CONJG(CX(I))*CY(I)



END DO
ELSE

*
* code for unequal increments or equal increments
* not equal to 1
*

IX = 1
IY = 1
IF (INCX.LT.0) IX = (-N+1)*INCX + 1
IF (INCY.LT.0) IY = (-N+1)*INCY + 1
DO I = 1,N

CTEMP = CTEMP + CONJG(CX(IX))*CY(IY)
IX = IX + INCX
IY = IY + INCY

END DO
END IF
CDOTC = CTEMP
RETURN

*
* End of CDOTC
*

END


